Implicit Differentiation

Objective This lab assignment provides practice with implicit differentiation.

Background Implicit differentiation requires the explicit identification of the independent variable. All other variables in the equation are either parameters (constants) or dependent variables (functions) of the independent variable. Once these dependencies are recognized, the Chain Rule is applied when differentiating both sides of the equation (with respect to an independent variable).

Two different ways to perform implicit differentiation in Maple will be presented. The difference in the approaches is the way the functional dependence is represented. One method closely resembles the steps that are used to implement implicit differentiation by hand (see Example 2). The more efficient method, demonstrated in Example 1, requires the use of a new Maple command.

New Maple commands introduced in this lab are `implicitdiff` and `allvalues`. As the name suggests, the `implicitdiff` command returns a derivative computed using implicit differentiation. The simplest usage is `implicitdiff(` *eq*`,` *y*`,` *x* `);` to determine $\frac{dy}{dx}$ where the equation *eq* implicitly defines y as a function of x. For example, the command

```
> implicitdiff( x^2+y^2=1, y, x );
```

returns $\frac{-x}{y}$ because $\frac{dy}{dx} = \frac{-x}{y}$ for all points on the unit circle (with $y \neq 0$). Higher-order derivatives are obtained by repeating the independent variable, exactly as is done for higher-order derivatives with `diff`.

Maple uses the `RootOf` notation to provide a generic representation of all solutions of certain equations that have multiple solutions. The `allvalues` command is used to force Maple to display all values represented by a `RootOf`. For example,

```
> q:=solve( { x+y=x*y, y-x=1 }, { x, y } );
> allvalues( q );
```

Note that if `solve` returns more than one solution involving `RootOf`, then the argument to `allvalues` should be made into a list, e.g., `allvalues( [q] );`.

Discussion Enter, and execute, the following Maple commands in a Maple worksheet.

Example 1: Implicit Differentiation with Implicit Dependence

```
> restart;                                        # clear Maple's memory
> with( plots );                                  # load package
> eq := (x^2+y^2-a*y)^2 = x^2+y^2;                # limaçon
> L := unapply( eq, a );                          # create function
> L(2);                                           # limaçon w/a = 2
> implicitplot( L(2), x=-2..2, y=-1..4,           # plot of typical limaçon
>               grid=[50,50] );
> DyDx1 := implicitdiff( L(2), y, x );            # compute dy/dx
> dy/dx = factor( DyDx1 );                        # display final result
> q1 := solve( { L(2), DyDx1=0}, {x,y} );         # pts on curve w/m = 0
> q2 := allvalues( [q1] );                        # remove RootOf
```

Example 2: Implicit Differentiation with Explicit Dependence

```
> eq2 := eval( L(2), y=y(x) );                    # limaçon w/expl dep
> Deq2 := diff( eq2, x );                         # diff eqn wrt x
> q3 := isolate( Deq2, diff(y(x),x) );            # solve for dy/dx
> DyDx2 := simplify( rhs( q3 ) );                 # simplify result
> dy/dx = DyDx2;                                  # display final result
```

Example 3: Higher-Order Implicit Derivatives

```
> L(2);                                       # recall specific limaçon
> dy/dx = DyDx1;                              # recall dy/dx
> DyDDx := implicitdiff( L(2), y, x,x );      # compute d^2y/dx^2
> DyDDDx := implicitdiff( L(2), y, x,x,x );   # compute d^3y/dx^3
```

Notes

(1) The `implicitplot` command is used to plot a curve when it is not possible, or not convenient, to write $y = f(x)$. The general syntax is `implicitplot(` *eqn*`, x=`*horizontalrange*`, y=`*verticalrange* `);`. If the resolution in this plot is poor, add the optional argument `grid=[`n_x`,`n_y`]` with reasonable values for n_x and n_y. The default grid is 25×25, for 625 points.

(2) When using implicit differentiation to find a derivative, $\frac{dy}{dx}$, the fundamental requirement is that all occurrences of the dependent variable are treated as functions of the independent variable, i.e., $y(x)$. The `implicitdiff` command takes care of this automatically by assuming the second and third arguments are the dependent and independent variables, respectively. (Additional arguments are used to indicate higher-order derivatives. See Example 3.)

(3) When `solve` finds more than one solution to an equation or system of equations, the answer often is expressed as the `RootOf` some auxiliary equation. While these expressions are often quite messy, do not assume they are of no use or interest. Sometimes the `allvalues` command can give exact values for the solutions. If approximate solutions are suitable, apply `evalf` to the result of `allvalues`.

(4) In general, a curve does not have a tangent line if neither $\frac{dy}{dx}$ nor $\frac{dx}{dy}$ exist at a point on the curve.

(5) Second-order and higher-order derivatives obtained by implicit differentiation can yield extremely complicated results. A tool like Maple is very useful for problems that do require these derivatives.

Questions

(1) Consider the curve defined implicitly by $x^2 - xy + 3y^2 = 6$.
  (a) Find the coordinates (x, y) of each x- and y-intercept of this curve.
  (b) Find $\frac{dy}{dx}$.
  (c) Find $\frac{d^2y}{dx^2}$. (Write your answer in factored form.)
  (d) Find all points (x, y) on this curve where $\frac{dy}{dx} = 0$.
  (e) Find the value of the second derivative at each point found in (d).

(2) This question refers to the curve introduced in Example 1.
  (a) Find the coordinates of the points on the limaçon with $a = 1$ where the tangent line is horizontal.
  (b) Find the coordinates of the points on the limaçon with $a = 1$ where the tangent line is vertical.
  (c) Find the coordinates of the points on the limaçon with $a = 1$ where the curve does not have a tangent line.

Extra Credit This question refers to the curve introduced in Example 1.
  (a) Find the English translation of the French word *limaçon*.
  (b) What is another name for a limaçon with $a = 0$?
  (c) Explain why a limaçon with $a = 1$ is called a cardioid.